<!--#Include File="Include.asp"--><%
'--------------------------------------
'Power By 方卡在线 @2010-2014 f02
'URL:http://www.fangka.net/
'EMail:admin@ifangka.com
'QQ:861118936
'版权所有 未经允许不得去除本版权信息
'--------------------------------------

Call CheckAdminType("0,1",UType)
Call CheckAdminLimit(ULimit,"m_Menu",0)

Dim Fk_Menu_Name,Fk_Menu_Dir,Fk_Menu_Template

Types=Clng(Request.QueryString("Type"))

Select Case Types
	Case 1
		Call MenuList() '菜单列表
	Case 2
		Call MenuAddForm() '添加菜单表单
	Case 3
		Call MenuAddDo() '执行添加菜单
	Case 4
		Call MenuEditForm() '修改菜单表单
	Case 5
		Call MenuEditDo() '执行修改菜单
	Case 6
		Call MenuDelDo() '执行删除菜单
	Case Else
		Call FKFun.ShowErr("未发现功能项！",1)
End Select

'==========================================
'函 数 名：MenuList()
'作    用：菜单列表
'==========================================
Sub MenuList()
%>
		<div id="rightNav">
			<ul>
				<li><a href="Menu.asp?Type=2" target="dialog" width="500">添加菜单</a></li>
			</ul>
		</div>
		<div id="searchLine" class="font_bold">
			菜单管理
		</div>
		<table width="100%" bordercolor="#CCCCCC" border="1" cellspacing="0" cellpadding="0">
			<tr>
				<th align="center">ID</th>
				<th align="center">名称</th>
				<th align="center">目录</th>
				<th align="center">模板方案</th>
				<th align="center">操作</th>
			</tr>
<%
	Sqlstr="Select Fk_Menu_Id,Fk_Menu_Name,Fk_Menu_Dir,Fk_Menu_Template From [Fk_Menu] Order By Fk_Menu_Id Asc"
	Rs.Open Sqlstr,Conn,1,1
	If Not Rs.Eof Then
		i=1
		While Not Rs.Eof
%>
			<tr>
				<td height="20" align="center"><%=Rs("Fk_Menu_Id")%></td>
				<td align="center"><%=Rs("Fk_Menu_Name")%></td>
				<td align="center"><%=Rs("Fk_Menu_Dir")%></td>
				<td align="center"><%=Rs("Fk_Menu_Template")%></td>
				<td align="center"><a href="Menu.asp?Type=4&Id=<%=Rs("Fk_Menu_Id")%>" target="dialog" width="500">修改</a> <a href="Menu.asp?Type=6&Id=<%=Rs("Fk_Menu_Id")%>" target="delete" confirmStr="您确认要删除“<%=Rs("Fk_Menu_Name")%>”？此操作不可逆！" refreshUrl="Menu.asp?Type=1" showStr="名称删除成功！">删除</a></td>
			</tr>
<%
			Rs.MoveNext
			i=i+1
		Wend
	Else
%>
			<tr>
				<td height="25" colspan="20" align="center">暂无记录</td>
			</tr>
<%
	End If
	Rs.Close
%>
			<tr>
				<td height="30" colspan="20">&nbsp;</td>
			</tr>
		</table>
		<div id="rightBottom">
		
		</div>
<%
End Sub

'==========================================
'函 数 名：MenuAddForm()
'作    用：添加菜单表单
'==========================================
Sub MenuAddForm()
%>
<form id="MenuAdd" name="MenuAdd" method="post" target="ajax" action="Menu.asp?Type=3" showStr="新菜单添加成功！" refreshUrl="Menu.asp?Type=1" boxClose="1" menuChange="1">
<div id="title">添加新菜单</div>
<div id="box">
	<table width="90%" border="1" bordercolor="#CCCCCC" align="center" cellpadding="0" cellspacing="0">
		<tr>
			<td width="24%" height="30" align="right">名称：</td>
			<td width="76%">&nbsp;<input name="Fk_Menu_Name" type="text" class="input" id="Fk_Menu_Name" /><span class="help" title="<p>菜单名称，请输入1-50个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">目录：</td>
			<td>&nbsp;<input name="Fk_Menu_Dir" type="text" class="input" id="Fk_Menu_Dir" /> *添加后不可修改<span class="help" title="<p>菜单目录，可留空，不可重复，添加后不可修改，请输入1-50个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">模板方案：</td>
            <td>&nbsp;<select name="Fk_Menu_Template" class="input" id="Fk_Menu_Template">
					<option value="">默认方案</option>
<%
	TempArr=Split(FKFun.GetFloderList("../skin/"&Fk_Site_Template&"/"),"||")
	For Each Temp In TempArr
		If Left(Temp,2)="t_" Then
%>
					<option value="<%=Temp%>"><%=Temp%></option>
<%
		End If
	Next
%>
				</select><span class="help" title="<p>菜单模板方案请放置在主方案下，目录以t_开头。</p>"></span>
			</td>
		</tr>
	</table>
</div>
<div id="bottom">
	<input type="submit" class="button" name="Enter" id="Enter" value="添 加" />
	<input type="button" onclick="dialogClose();" class="button" name="button" id="button" value="关 闭" />
</div>
</form>
<%
End Sub

'==============================
'函 数 名：MenuAddDo
'作    用：执行添加菜单
'==============================
Sub MenuAddDo()
	Fk_Menu_Name=FKFun.HTMLEncode(Trim(Request.Form("Fk_Menu_Name")))
	Fk_Menu_Dir=Server.URLEncode(FKFun.HTMLEncode(Trim(Request.Form("Fk_Menu_Dir"))))
	Fk_Menu_Template=FKFun.HTMLEncode(Trim(Request.Form("Fk_Menu_Template")))
	Call FKFun.JsonString(Fk_Menu_Name,1,50,0,"请输入名称！","名称不能大于50个字符！")
	Call FKFun.JsonString(Fk_Menu_Dir,1,50,2,"请输入目录！","目录不能大于50个字符！")
	Call FKFun.JsonString(Fk_Menu_Template,1,50,2,"请选择模板方案！","模板方案不能大于50个字符！")
	Sqlstr="Select Fk_Menu_Id,Fk_Menu_Name,Fk_Menu_Dir,Fk_Menu_Template From [Fk_Menu] Where Fk_Menu_Name='"&Fk_Menu_Name&"'"
	Rs.Open Sqlstr,Conn,1,3
	If Rs.Eof Then
		Application.Lock()
		Rs.AddNew()
		Rs("Fk_Menu_Name")=Fk_Menu_Name
		Rs("Fk_Menu_Dir")=Fk_Menu_Dir
		Rs("Fk_Menu_Template")=Fk_Menu_Template
		Rs.Update()
		Application.UnLock()
		Call FKFun.JsonSuccess()
	Else
		Rs.Close
		Call FKFun.JsonError("该菜单名称已经被占用！")
	End If
	Rs.Close
End Sub

'==========================================
'函 数 名：MenuEditForm()
'作    用：修改菜单表单
'==========================================
Sub MenuEditForm()
	Id=Clng(Request.QueryString("Id"))
	Sqlstr="Select Fk_Menu_Name,Fk_Menu_Dir,Fk_Menu_Template From [Fk_Menu] Where Fk_Menu_Id=" & Id
	Rs.Open Sqlstr,Conn,1,1
	If Not Rs.Eof Then
		Fk_Menu_Name=Rs("Fk_Menu_Name")
		Fk_Menu_Dir=Rs("Fk_Menu_Dir")
		Fk_Menu_Template=Rs("Fk_Menu_Template")
	Else
		Rs.Close
		Call FKFun.ShowErr("菜单不存在！",1)
	End If
	Rs.Close
%>
<form id="MenuEdit" name="MenuEdit" method="post" target="ajax" action="Menu.asp?Type=5" showStr="菜单修改成功！" refreshUrl="Menu.asp?Type=1" boxClose="1" menuChange="1">
<div id="title">修改菜单</div>
<div id="box">
	<table width="90%" border="1" bordercolor="#CCCCCC" align="center" cellpadding="0" cellspacing="0">
		<tr>
			<td width="24%" height="30" align="right">名称：</td>
			<td width="76%">&nbsp;<input name="Fk_Menu_Name" type="text" class="input" id="Fk_Menu_Name" value="<%=Fk_Menu_Name%>" /><span class="help" title="<p>菜单名称，请输入1-50个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">目录：</td>
			<td>&nbsp;<input name="Fk_Menu_Dir" type="text" class="input" id="Fk_Menu_Dir" value="<%=Fk_Menu_Dir%>" readonly="readonly" /><span class="help" title="<p>菜单目录，可留空，不可重复，添加后不可修改，请输入1-50个字符（两个字符为一个汉字）。</p>"></span></td>
		</tr>
		<tr>
			<td height="30" align="right">模板方案：</td>
            <td>&nbsp;<select name="Fk_Menu_Template" class="input" id="Fk_Menu_Template">
					<option value="">默认方案</option>
<%
	TempArr=Split(FKFun.GetFloderList("../skin/"&Fk_Site_Template&"/"),"||")
	For Each Temp In TempArr
		If Left(Temp,2)="t_" Then
%>
					<option value="<%=Temp%>"<%=FKFun.BeSelect(Fk_Menu_Template,Temp)%>><%=Temp%></option>
<%
		End If
	Next
%>
				</select><span class="help" title="<p>菜单模板方案请放置在主方案下，目录以t_开头。</p>"></span>
			</td>
		</tr>
	</table>
</div>
<div id="bottom">
	<input type="hidden" name="Id" value="<%=Id%>" />
	<input type="submit" class="button" name="Enter" id="Enter" value="修 改" />
	<input type="button" onclick="dialogClose();" class="button" name="button" id="button" value="关 闭" />
</div>
</form>
<%
End Sub

'==============================
'函 数 名：MenuEditDo
'作    用：执行修改菜单
'==============================
Sub MenuEditDo()
	Fk_Menu_Name=FKFun.HTMLEncode(Trim(Request.Form("Fk_Menu_Name")))
	Fk_Menu_Dir=Server.URLEncode(FKFun.HTMLEncode(Trim(Request.Form("Fk_Menu_Dir"))))
	Fk_Menu_Template=FKFun.HTMLEncode(Trim(Request.Form("Fk_Menu_Template")))
	Id=Trim(Request.Form("Id"))
	Call FKFun.JsonString(Fk_Menu_Name,1,50,0,"请输入名称！","名称不能大于50个字符！")
	Call FKFun.JsonString(Fk_Menu_Dir,1,50,2,"请输入目录！","目录不能大于50个字符！")
	Call FKFun.JsonString(Fk_Menu_Template,1,50,2,"请选择模板方案！","模板方案不能大于50个字符！")
	Call FKFun.JsonNum(Id,"系统参数错误，请刷新页面！")
	Sqlstr="Select Fk_Menu_Id,Fk_Menu_Name,Fk_Menu_Dir,Fk_Menu_Template From [Fk_Menu] Where Fk_Menu_Id=" & Id
	Rs.Open Sqlstr,Conn,1,3
	If Not Rs.Eof Then
		Application.Lock()
		Rs("Fk_Menu_Name")=Fk_Menu_Name
		Rs("Fk_Menu_Dir")=Fk_Menu_Dir
		Rs("Fk_Menu_Template")=Fk_Menu_Template
		Rs.Update()
		Application.UnLock()
		Call FKFun.JsonSuccess()
	Else
		Rs.Close
		Call FKFun.JsonError("要修改的菜单不存在！")
	End If
	Rs.Close
End Sub

'==============================
'函 数 名：MenuDelDo
'作    用：执行删除菜单
'==============================
Sub MenuDelDo()
	Id=Trim(Request.QueryString("Id"))
	Call FKFun.ShowNum(Id,"系统参数错误，请刷新页面！")
	Sqlstr="Select Fk_Module_Id From [Fk_Module] Where Fk_Module_Menu=" & Id
	Rs.Open Sqlstr,Conn,1,1
	If Not Rs.Eof Then
		Call FKFun.ShowErr("该菜单有菜单项，无法删除！",2)
	End If
	Rs.Close
	Sqlstr="Select Fk_Menu_Id From [Fk_Menu] Where Fk_Menu_Id=" & Id
	Rs.Open Sqlstr,Conn,1,3
	If Not Rs.Eof Then
		Application.Lock()
		Rs.Delete()
		Application.UnLock()
		Response.Write("ok")
	Else
		Response.Write("要删除的菜单不存在！")
	End If
	Rs.Close
End Sub

Call FKFun.PageEnd()
%>
